Diferenciální rovnice I

V kurzu Diferenciální rovnice I se naučíme pomocí počítačového algebraického systému Maple hledat obecná a partikulární řešení obyčejných diferenciálních rovnic 1. řádu. Pro případ, že nelze partikulární řešení nalézt v analytické tvaru (nebo systém Maple neumí řešení nalézt), naučíme se řešit diferenciální rovnice numericky. Upozorníme na ,,zrádnosti" systému Maple, které odhalíme a se kterými si rovněž poradíme, ovládáme-li dobře látku předmětu Matematika I a II (kapitola Diferenciální rovnice).

Poznámka. Budeme se snažit o přehledný (strukturovaný) zápis příkazů v systému Maple. Na začátku si vysvětlíme použití příkazů, které jsou nezbytné pro řešení všech příkladů tohoto kurzu. Pak předvedeme vzorová řešení standardních úloh pomocí systému Maple. Dříve než začneme úlohu řešit, seznámíme se s novými příkazy. Postup řešení je komentován na konci příkladu.

Symbolicky diferenciální rovnici 1. řádu pro neznámou funkci $y = y(x)$ píšeme ve tvaru

$$F(x, y, y') = 0\,.$$

Obecné řešení pro většinu rovnic neumíme najít. Zda systém Maple nalezne řešení, se přesvědčíme pomocí příkazu `dsolve(DR,y(x))`, kde první parametr `DR` je daná diferenciální rovnice a druhý `y(x)` je název pro hledanou funkci. Připomeňme si některé dovednosti, které jsme mohli získat v kurzu Derivace nebo Řešení rovnic. V systému Maple derivaci funkce y podle x zapíšeme příkazem `diff(y(x),x)`. Je-li pravá strana rovnice rovna nule, nemusíme ji psát. Jinak řečeno, napíšeme-li místo rovnice pouze analytický výraz, systém Maple za pravou stranu rovnice dosadí nulu.

Příklad 1. Nalezněte obecné řešení nehomogenní lineární diferenciální rovnice s konstantními koeficienty

$$y' + y = xe^{-x} + 1\,.$$

```
> DR:=diff(y(x),x)+y(x)=x*exp(-x)+1;
```

$$DR := \left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right) + y(x) = x e^{(-x)} + 1$$

```
> dsolve(DR,y(x));
```

$$y(x) = \left(\frac{x^2}{2} + e^x + \_C1\right) e^{(-x)}$$

Příklad 2. Nalezněte obecné řešení nehomogenní lineární diferenciální rovnice

$$xy' + 2y = \sin x\,.$$

```
> DR:=x*diff(y(x),x)+2*y(x)=sin(x);
```

$$DR := x\left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right) + 2y(x) = \sin(x)$$

```
> dsolve(DR,y(x));
```

$$y(x) = \frac{\sin(x) - x\cos(x) + \_C1}{x^2}$$

Příklad 3. Nalezněte obecné řešení nelineární diferenciální rovnice

$$y' + y^2 \sin x = 0\,.$$

```
> DR:=diff(y(x),x)+(y(x))^2*sin(x);
```

$$DR := \left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right) + y(x)^2 \sin x$$

```
> dsolve(DR,y(x));
```

$$y(x) = \frac{1}{-\cos(x) + \_C1}$$

Poznámka. Systém Maple označuje integrační konstantu symbolem $\_C1$. Předchozí příklady slouží jako ukázka rychlého integrování diferenciálních rovnic. Jde pouze o formální nalezení analytického výrazu, pomocí kterého se dá řešení vyjádřit. Ovšem k funkčnímu předpisu potřebujeme znát definiční obor. Bez něj není úloha správně vyřešena. V tomto směru nám systém Maple nepomůže. Jak určit definiční obor se dozvíme ve skriptech Matematika I a II.

Příklad 4. Nalezněte obecné řešení nehomogenní lineární diferenciální rovnice

$$xy' + 2y = x^2 - x + 1$$

a partikulární řešení, které vyhovuje počáteční podmínce $y(1) = \frac{1}{2}$. (Součástí úlohy je stanovení definičního oboru nalezeného řešení.) Popište, jak se řešení chová pro $x \to \infty$. Nakreslete integrální křivku, která prochází bodem $(1, \frac{1}{2})$.

Nápověda. Vysvětlíme si nejprve použití příkazů, které následně využijeme při řešení zadané úlohy. S některými jsme se mohli setkat v jiných kurzech, přesto je připomeneme. Chceme-li najít řešení počáteční úlohy, pak použijeme příkaz `dsolve({DR,PP},y(x));`, kde parametr `PP` je počáteční podmínka a význam ostatních parametrů již známe. Složená závorka je nezbytná. Příkaz `dsolve` vrací řešení ve tvaru rovnice, na jejíž pravé straně se obvykle nachází funkční předpis daného řešení. Potřebujeme-li získat pravou stranu rovnice, použijeme příkaz `rhs` (right-hand side). Protože budeme s řešením pracovat jako s funkcí, měli bychom vědět jak z analytického výrazu vytvořit funkční předpis. K tomu slouží příkaz `unapply(vyraz,x)`, který z výrazu `vyraz` udělá funkci proměnné `x`. O chování nalezeného řešení v nekonečnu vypovídá limita řešení pro $x \to \infty$, tj. použijeme příkaz `limit(f(x),x=infinity);`. Graf funkce f na intervalu $\langle a, b\rangle$ nakreslíme pomocí příkazu `plot(f(x),x=a..b);`.

```
> DR:=x*diff(y(x),x)+2*y(x)=x^2-x+1;
```

$$DR := x\left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right) + 2\,y(x) = x^2 - x + 1$$

```
> dsolve(DR,y(x));
```

$$y(x) = \frac{x^2}{4} - \frac{x}{3} + \frac{1}{2} + \frac{\_C1}{x^2}$$

```
> PP:=y(1)=1/2;
```

$$PP := y(1) = \frac{1}{2}$$

```
> res:=dsolve({DR,PP},y(x));
```

$$res := y(x) = \frac{x^2}{4} - \frac{x}{3} + \frac{1}{2} + \frac{1}{12x^2}$$

```
> f:=unapply(rhs(factor(res)),x);
```

$$f := x \to \frac{1}{12}\,\frac{3x^4 - 4x^3 + 6x^2 + 1}{x^2}$$

```
> f(1);
```

$$\frac{1}{2}$$

```
> limit(f(x),x=infinity);
```

$$\infty$$

```
> limit(f(x),x=0);
```

$$\infty$$

```
> plot(f(x),x=0.5..2);
```



Z přednášek MI víme, jak určit definiční obor řešení. Protože koeficienty $a_0(x) = x$, $a_1(x) = 2$ a pravá strana $b(x) = x^2 - x + 1$ jsou spojité funkce na $\mathbb{R}$ a protože $a_0(x) \neq 0$ pro $x \neq 0$, je definičním oborem řešení $y(x) = \frac{3x^4 - 4x^3 + 6x^2 + C_1}{12x^2}$ buď interval $(-\infty, 0)$ nebo interval $(0, \infty)$. Zdůrazněme, že definiční obor řešení musíme určit na základě vědomostí z teorie lineárních diferenciálních rovnic. Protože počáteční podmínka je zadaná pro $x = 1$, je definiční obor řešení $y(x) = \frac{3x^4 - 4x^3 + 6x^2 + 1}{12x^2}$ interval $(0, \infty)$. Řešení roste nade všechny meze pro $x \to \infty$.

Poznámka. Limitu řešení v nule jsme spočetli pro lepší porozumění obrázku integrální křivky. Příkazem `factor` jsme docílili funkčního předpisu ve tvaru jednoho zlomku.

Příklad 5. Nalezněte řešení nelineární diferenciální rovnice

$$y' = \frac{1+2x}{2y} \tag{1}$$

které vyhovuje počáteční podmínce a) $y(0)=0$, b) $y(0)=\frac{1}{4}$, c) $y(0)=\frac{1}{2}$, d) $y(0)=1$.

Nápověda. Úkolem je řešit jednu diferenciální rovnici pro čtyři různé počáteční podmínky. Chceme-li se vyhnout opakování postupu, můžeme napsat cyklus nebo alespoň počáteční podmínky umístit do seznamu a odvolávat se na jejich pořadí.

```
> DR:=diff(y(x),x)=(1+2*x)/(2*y(x));
```

$$DR := \frac{\mathrm{d}}{\mathrm{d}x}y(x) = \frac{1}{2}\frac{1+2x}{y(x)}$$

```
> PP:=[y(0)=0,y(0)=1/4,y(0)=1/2,y(0)=1];
```

$$PP := \left[y(0)=0, y(0)=\frac{1}{4}, y(0)=\frac{1}{2}, y(0)=1\right]$$

```
> dsolve({DR,PP[1]},y(x));
```

$$y(x)=\sqrt{x+x^2}, y(x)=-\sqrt{x+x^2}$$

```
> dsolve({DR,PP[2]},y(x));
```

$$y(x)=\sqrt{x+x^2+\frac{1}{16}}$$

```
> dsolve({DR,PP[3]},y(x));
```

$$y(x) = \sqrt{x + x^2 + \frac{1}{4}}$$

```
> dsolve({DR,PP[4]},y(x));
```

$$y(x) = \sqrt{x + x^2 + 1}$$

Cyklus je pouze technicky jiný postup. Jeho struktura by měla být zřejmá, proto ji nevysvětlujeme. Cílem je upozornit na tuto možnost.

```
> for X in PP do
  res:=dsolve({DR,X},y(x));
  print(X);
  print(res);
  od:
```

$$y(0) = 0$$

$$y(x) = \sqrt{x + x^2}, y(x) = -\sqrt{x + x^2}$$

$$y(0) = \frac{1}{4}$$

$$y(x) = \sqrt{x + x^2 + \frac{1}{16}}$$

$$y(0) = \frac{1}{2}$$

$$y(x) = \sqrt{x + x^2 + \frac{1}{4}}$$

$$y(0) = 1$$

$$y(x) = \sqrt{x + x^2 + 1}$$

Poznámka. Povšimněme si, že první počáteční podmínka nemá pro diferenciální rovnici (1) smysl. Zlomek na pravé straně rovnice (1) je definován pro $y \neq 0$. Přesto systém Maple našel řešení a hned dvě. Ta ale řeší počáteční úlohu $2yy' = 1 + 2x$, $y(0) = 0$. Vidíme, že systém Maple musí být pod stálou kontrolou uživatele.

Nesmíme zapomenout určit definiční obor partikulárních řešení. Podívejme se na problém zevrubněji. Obecné řešení diferenciální rovnice (1) umíme najít metodou separace proměnných. Pokud najdeme příslušné primitivní funkce, metoda dává rovnici pro neznámou x a y, ze které se někdy podaří y osamostatnit. V našem případě požádáme systém Maple, aby řešení nehledal v analytickém tvaru, ale dal nám rovnici pro neznámou x a y. V příkazu `dsolve` použijeme volbu `implicit`.

```
> dsolve(DR,y(x),implicit);
```

$$y(x)^2 - x - x^2 - \_C1 = 0$$

V předmětu Matematika I jsme se naučili, že integrální křivka řešení diferenciální rovnice (1) bude ležet buď v obdélníku $(-\infty, \infty) \times (-\infty, 0)$ nebo v obdélníku $(-\infty, \infty) \times (0, +\infty)$ a že probíhá daný obdélník „od hranice k hranici“. Rovnice $y^2(x) - x - x^2 - c_1 = 0$ má smysl, jestliže $x^2 + x + c_1 > 0$. Víme, že je-li $c_1 > \frac{1}{4}$, pak kvadratický trojčlen $x^2 + x + c_1$ nabývá pro každé $x \in \mathbb{R}$ kladných hodnot, neboť je jeho diskriminant záporný a koeficient v kvadratickém členu je kladný. V případě, že $c_1 \leq \frac{1}{4}$, je nerovnost splněna pro $x \in (-\infty, -\frac{1}{2} - \frac{\sqrt{1-4c_1}}{2}) \cup (-\frac{1}{2} + \frac{\sqrt{1-4c_1}}{2}, +\infty)$. Kořeny kvadratického trojčlenu umí systém Maple nalézt, viz Řešení rovnic.

```
> solve(x^2+x+_C1,x);
```

$$-\frac{1}{2} + \frac{\sqrt{1-4\_C1}}{2}, -\frac{1}{2} - \frac{\sqrt{1-4\_C1}}{2}$$

Poznámka. Pozor! Řešením diferenciální rovnice je funkce jistých vlastností, jejímž definičním oborem je právě jeden interval. Tedy v případě, že $c_1 \leq \frac{1}{4}$, musíme vybrat jeden interval ze sjednocení $(-\infty, -\frac{1}{2} - \frac{\sqrt{1-4c_1}}{2}) \cup (-\frac{1}{2} + \frac{\sqrt{1-4c_1}}{2}, +\infty)$ v závislosti na počáteční podmínce.

Naše úvaha bude zřejmá, uvědomíme-li si, že rovnice $y^2 - x^2 - x - c_1 = 0$ je pro $c_1 \neq \frac{1}{4}$ rovnicí hyperboly se středem $(-\frac{1}{2}, 0)$, neboť rovnici lze upravit na tvar

$$\frac{y^2}{c_1 - \frac{1}{4}} - \frac{(x+\frac{1}{2})^2}{c_1 - \frac{1}{4}} = 1\,.$$

Je-li $c_1 - \frac{1}{4} < 0$, pak je hyperbola souměrná podle osy x a v případě, že $c_1 - \frac{1}{4} > 0$, pak je souměrná podle přímky $x = -\frac{1}{2}$. Pro $c_1 = \frac{1}{4}$ zřejmé úpravy vedou na tvar $|y| = |x + \frac{1}{2}|$, tj. rovnici dvou různoběžných přímek – asymptot hyperboly. Křivky v následujícím obrázku odpovídají hodnotám $c_1 = \frac{1}{16}, \frac{1}{4}, 1$. Povšimněme si, že celá větev zelené hyperboly je grafem funkce proměnné x, zatímco pro černou hyperbolu toto neplatí.



Dosazení příslušných konstant ($c_1 = \frac{1}{16}, \frac{1}{4}, 1$) nás vede k závěru, že v případě b) je definičním oborem partikulárního řešení interval $(\frac{\sqrt{3}-2}{4}, +\infty)$, v případě c) $(-\frac{1}{2}, +\infty)$ a v případě d) $(-\infty, +\infty)$. Příslušné integrální křivky můžeme nahlédnout v předchozím obrázku. Rozmysleme si dobře, jak vypadají.

Poznamenejme, že integrální křivky můžeme nakreslit, aniž bychom řešili zadanou diferenciální rovnici. K tomu slouží příkaz

$$\mathtt{DEplot}([\mathtt{DR}], [\mathtt{y(x)}], \mathtt{x} = \mathtt{a..b}, [[\mathtt{PP1}], [\mathtt{PP2}], [\mathtt{PP3}], \ldots], \mathtt{volba});,$$

kde `DR` je daná diferenciální rovnice, `y(x)` je symbol pro řešení, `x=a..b` je rozmezí nezávisle proměnné a `PP1, PP2, PP3` jsou počáteční podmínky. Každá počáteční podmínka dá vzniknout jedné integrální křivce. Pozor! Příkaz `DEplot` je součástí souboru programů `DEtools`. Buď tento soubor programů před použitím příkazu `DEplot` přivoláme příkazem `with(DEtools):` nebo musíme užít komplikovanější tvar `DEtools[DEplot](...);`. Vraťme se k Příkladu 5 a nakresleme integrální křivky řešení pro varianty b) – d). Využijeme volitelný parametr `volba` pro vymezení y–ové souřadnice a zvolíme si barvy křivek.

```
> DR:=diff(y(x),x)=(1+2*x)/(2*y(x));
```

$$DR := \frac{\mathrm{d}}{\mathrm{d}x}y(x) = \frac{1}{2}\frac{1+2x}{y(x)}$$

```
> DEtools[DEplot]([DR],[y(x)],x=0..2,[[y(0)=1/4],[y(0)=1/2],[y(0)=1]],
  y=-1..3,linecolor=[black,blue,green]);
```



Rovnice, jejichž řešení hledáme pomocí metody separace proměnných, často nesplňují předpoklady věty o jednoznačnosti řešení. V tomto případě je zapotřebí systém Maple řádně kontrolovat, neboť úlohu nevyřeší korektně. Následující příklad ilustruje tuto zrádnost.

Příklad 6. Nalezněte řešení počáteční úlohy

$$y' = -3\sqrt[3]{y^2}\,, \quad y(0) = 8 \tag{2}$$

a nakreslete jeho graf.

```
> DR:=diff(y(x),x)=-3*(y(x))^(2/3);
```

$$DR := \frac{\mathrm{d}}{\mathrm{d}x}y(x) = -3\,y(x)^{\left(\frac{2}{3}\right)}$$

```
> PP:=y(0)=8;
```

$$PP := y(0) = 8$$

```
> pres:=dsolve({DR,PP},y(x));
```

$$pres := y(x) = \mathrm{RootOf}\left(\_Z^{\left(\frac{1}{3}\right)} + x - 8^{\left(\frac{1}{3}\right)}\right)$$

```
> f:=unapply(rhs(pres),x);
```

$$f := x \to \mathrm{RootOf}\left(\_Z^{\left(\frac{1}{3}\right)} + x - 8^{\left(\frac{1}{3}\right)}\right)$$

> `plot(f(x),x=0..3);`



Poznámka. Symbol `RootOf(vyraz)` v systému Maple reprezentuje všechny kořeny rovnice $\mathrm{vyraz} = 0$ pro neznámou $\_Z$.

Vraťme se k diferenciální rovnici (2). Pravá strana rovnice $-3\sqrt[3]{y^2}$ je funkce spojitá na $\mathbb{R} \times \mathbb{R}$ (funkční předpis nezávisí na proměnné x), tedy řešení prochází od „hranice k hranici" obdélníka $\mathbb{R} \times \mathbb{R}$.

V předchozím obrázku integrální křivka skončila v bodě $(2,0)$ a to není možné. Zřejmě $y(x) \equiv 0\,, x \in \mathbb{R}$ je konstantní řešení rovnice (2). Zdá se, že partikulární řešení můžeme prodloužit konstantní nulou. Systém Maple říká, že řešení počáteční úlohy (2) splňuje algebraickou rovnici

$$y^{\frac{1}{3}} + x - 2 = 0\,.$$

Ke stejnému závěru dospějeme, použijeme-li metodu separace proměnných pro $y \neq 0$. Vzhledem k počáteční podmínce je $y > 0$. Tedy $y = (-x+2)^3$ za předpokladu, že $-x+2 > 0$, tj. $x < 2$. Nabízí se otázka, zda funkce $y(x) = (-x+2)^3\,, x \in \mathbb{R}$ není řešením úlohy (2). Spočtěme $y'(x)$

$$y'(x) = -3(-x+2)^2\,, \quad x \in \mathbb{R}\,.$$

Dosadíme do pravé strany rovnice (2) za y

$$-3[y(x)]^{\frac{2}{3}} = -3\left[(-x+2)^3\right]^{\frac{2}{3}} = -3(-x+2)^2\,, \quad x \in \mathbb{R}\,.$$

Ověřili jsme, že funkce $y(x) = (-x+2)^3\,, x \in \mathbb{R}$ je diferencovatelnou funkcí na $\mathbb{R}$ a že splňuje rovnici (2), tedy je řešením. Ukážeme, že řešení, které systém Maple našel, lze pro $x > 2$ také prodloužit konstantní nulou. Neboť $y(2) = 0$ a $y'(2) = 0$. Funkce

$$y(x) = \begin{cases} (-x+2)^3\,, & x \leq 2 \\ 0\,, & x > 2 \end{cases}$$

je spojitá a diferencovatelná na $\mathbb{R}$ a vyhovuje diferenciální rovnici (2). Pomocí systému Maple nakreslíme grafy těchto dvou řešení.

Nápověda. Příkaz `piecewise(podm1,f1,podm2,f2,f)` nám umožňuje definovat funkci po částech. Je-li splněna podmínka `podm1`, platí funkční předpis `f1`, je-li splněna podmínka `podm2`, platí funkční předpis f2, jinak platí funkční předpis f.

```
> g:=x->(-x+2)^3;
```

$$g := x \to (-x+2)^3$$

```
> plot(g(x),x=0..4);
```




```
> f:=x->piecewise(x<=2,(-x+2)^3,0);
```

$$f := x \to \mathrm{piecewise}(x \le 2, (-x+2)^3, 0)$$

```
> plot(f(x),x=0..4,color=blue,thickness=5);
```



Další zkoumání by ukázalo, že Příklad 6 má nekonečně mnoho řešení, která mají tvar

$$y(x) = \begin{cases} (-x+2)^3, & x < 2 \\ \quad 0 \quad\quad\;\, , & 2 \le x \le c \\ (-x+c)^3, & c < x \end{cases}$$

kde $c \geq 2$. Pro hodnoty $c = 2,\ 3.5,\ \infty$ jsou řešení zakresleny v následujícím obrázku.



Všechna řešení jsme našli pomocí metody „slepování řešení". Studenty, kteří se zajímají o problematiku diferenciálních rovnic a chtěli by se seznámit s metodou „slepování řešení", odkazuji na knihu *Přehled užité matematiky II*, Karel Rektorys, str. 6–7.

Většina diferenciálních rovnic zejména nelineárních nemá analytické řešení. V případě, že jsou pro počáteční úlohu splněny předpoklady věty o existenci a jednoznačnosti řešení a řešení neumíme analyticky vyjádřit, alespoň můžeme spočítat přibližné řešení. K tomu slouží metody numerické matematiky. Systém Maple v případě, že řešení neumí nalézt, nedá žádnou odpověď. Můžeme alespoň nakreslit směrové pole. Se systémem Maple je to jednoduché. Použijeme příkaz `dfieldplot(DR,y(x),x=a..b,y=c..d,volba);` .

Tento příkaz stejně jako `DEplot` je součástí souboru programů `DEtools`.

Příklad 7. Nakreslete směrové pole diferenciální rovnice

$$y' = \frac{4 - xy}{1 + y^2}\,. \tag{3}$$

```
> DR:=diff(y(x),x)=(4-x*y(x))/(1+y(x)^2);
```

$$DR := \frac{\mathrm{d}}{\mathrm{d}x}y(x) = \frac{4 - xy(x)}{1 + y(x)^2}$$

```
> dfieldplot(DR,y(x),x=-10..10,y=-10..10,dirgrid=[40,40]);
```



Poznámka. Využili jsme volitelný parametr `dirgrid=[m,n]`, pomocí kterého jsme nastavili počet bodů sítě. Směrové pole se tak skládá z $m \times n$ segmentů.

Sami vyzkoušejte, že systém Maple nenajde obecné řešení diferenciální rovnice (3). Předchozí obrázek nám však dává jistou představu o řešení na intervalu $(-10, 10)$. Protože funkce $f(x, y) = \frac{4-xy}{1+y^2}$ je spojitá v $\mathbb{R}^2$, leží integrální křivky rovnice (3) v obdélníku $\mathbb{R} \times \mathbb{R}$ a přitom ho probíhají „od hranice k hranici". V každém svém bodě mají tečnu určenou směrovým polem diferenciální rovnice. Ověříme-li předpoklady věty o existenci a jednoznačnosti řešení, zjistíme, že rovnice (3) má pro každou počáteční podmínku právě jedno řešení. Má tudíž smysl, zajímat se o přibližné řešení.

Příklad 8. Určete přibližné hodnoty řešení počáteční úlohy

$$y' = \frac{4 - xy}{1 + y^2}, \quad y(0) = -2$$

pro $x = -1, -0.5, 0, 0.5, 1, 1.5, 2, 2.5, 3, 3.5, 4$ a vyneste je do roviny xy.

Nápověda. Hodnoty můžeme nalézt najednou pomocí příkazu

$$\mathtt{dsolve(\{DR, PP\}, y(x), numeric, output = array([x1, x2, ..., xk]));}.$$

Volitelný parametr `output=array([x1,x2,...])` zajistí výpis tabulky uspořádaných dvojic (x_i, y_i), $i = 1, 2, \ldots k$, kde y_i je přibližná hodnota řešení v x_i. Poznamenejme navíc, že zadané hodnoty x jsou členy posloupnosti, která je určena vzorcem pro n–tý člen $a_n = -1 + (n-1)\frac{1}{2}$. Hodnoty $x = -1, -0.5, 0, 0.5, 1, 1.5, 2, 2.5, 3, 3.5, 4$ můžeme pomocí systému Maple generovat příkazem `seq(-1+(n-1)*0.5,`
Příkaz, který vykreslí uspořádané dvojice do roviny xy, má strukturu `odeplot(tab,style=point);`, kde `tab` je tabulka uspořádaných dvojic získaná již zmíněným postupem. Příkaz je součástí souboru

programů `plots`.

```
> DR:=diff(y(x),x)=(4-x*y(x))/(1+y(x)^2);
```

$$DR := \frac{\mathrm{d}}{\mathrm{d}x}y(x) = \frac{4 - xy(x)}{1 + y(x)^2}$$

```
> PP:=y(0)=-2;
```

$$PP := y(0) = -2$$

```
> tab:=dsolve({DR,PP},y(x),numeric, output=array([seq(-1.0+(i-1)*0.5,
  i=1..11)]));
```

$$
tab := \begin{bmatrix} [x, y(x)] \\ \begin{bmatrix} -1.0 & -2.46916883175438384 \\ -0.500000000000000000 & -2.30536062864656133 \\ 0.0 & -2.0 \\ 0.500000000000000000 & -1.44763415200380452 \\ 1.0 & -0.143413274317184703 \\ 1.50000000000000000 & 1.06095212434664820 \\ 2.0 & 1.41011727792797204 \\ 2.50000000000000000 & 1.51699121804150505 \\ 3.0 & 1.49235107072983952 \\ 3.50000000000000000 & 1.38090099481315164 \\ 4.0 & 1.21806005488714053 \end{bmatrix} \end{bmatrix}
$$

```
> plots[odeplot](tab,style=point);
```



Užitím příkazu `dsolve({DR,PP},y(x),numeric);` najdeme přibližné řešení počáteční úlohy `DR,PP`. Označíme-li si toto řešení, pak snadno najdeme hodnotu řešení pro libovolné x. Vraťme se k Příkladu 8. Pro $x = 0.1$ hledejme přibližnou hodnotu řešení.

```
> res:=dsolve({DR,PP},y(x),numeric);
```

$$res := \textbf{proc}(x\_rkf45) \ \ldots \ \textbf{end proc}$$

```
> res(0.1);
```

$$[x = 0.1, y(x) = -1.91522136344974037]$$

Se systémem Maple můžeme nejenom vykreslit konkrétní uspořádané dvojice grafu přibližného řešení, ale také můžeme nakreslit celý graf. Stačí použít příkaz

$$\texttt{odeplot}(\texttt{res}, [\texttt{x}, \texttt{y}(\texttt{x})], \texttt{x} = \texttt{a..b}, \texttt{volba});,$$

kde `res` je výstup příkazu `dsolve({DR,PP},y(x),numeric);`. Nakresleme přibližné řešení počáteční úlohy z Příkladu 8.

```
> plots[odeplot](res,[x,y(x)],x=-20..20,numpoints=100);
```



Poznámka. Volitelný parametr `numpoints` nastavuje minimální počet bodů grafu. V našem případě jsme docílili vyhlazení křivky.

Poznámka. Systém Maple implicitně používá numerickou metodu Runge – Kutta Fehlberg. Typ metody lze do jisté míry ovlivnit, viz Help.